Topological Insulators 2 (Topological superconductors)
2020 Spring, Lecture 4
Majorana zero modes and robust ground-state degeneracy of the Kitaev chain

(Dated: March 10, 2020)

Last week, we have discussed simple quantum information protocols using a Kitaev double dot, a minimal model for 1D topological superconductors. In this lecture, we extend the discussion to larger system sizes and the presence of disorder. We discuss that the ground state of a long Kitaev chain is almost degenerate, even in the presence of disorder. This protected degeneracy suggests that these quantum states might be used for reliable storage of quantum information.

CONTRIBUTORS

Andras Palyi (writeup), Peter Boross (numerics, figures), Janos Asboth, Peter Boross, Laszlo Oroszlany (discussions).

I. A LONG, DISORDERED TOPOLOGICAL KITAEV CHAIN HAS A TWOFOLD DEGENERATE GROUND STATE

1. Recall (see Lecture 2) that the Kitaev double dot has the following BdG Hamiltonian in the 'local ordering' (1e, 1h, 2e, 2h):

$$\mathcal{H} = \begin{pmatrix} -\mu & 0 & v & \Delta \\ 0 & \mu & -\Delta & -v \\ v & -\Delta & -\mu & 0 \\ \Delta & -v & 0 & \mu \end{pmatrix}. \tag{1}$$

2. From this, it is straightforward to see that the BdG Hamiltonian of a half-infinite Kitaev chain in the topological fully dimerized limit ($v = \Delta = 1$, $\mu = 0$) reads

$$\mathcal{H} = \left(\begin{array}{cc|cc|cc|c} 0 & 0 & 1 & 1 & 0 & 0 & \dots \\ 0 & 0 & -1 & -1 & 0 & 0 & \dots \\ \hline 1 & -1 & 0 & 0 & 1 & 1 & \dots \\ 1 & -1 & 0 & 0 & -1 & -1 & \dots \\ \hline 0 & 0 & 1 & -1 & 0 & 0 & \dots \\ 0 & 0 & 1 & -1 & 0 & 0 & \dots \\ \hline \dots & & \dots & & \dots & & \dots \end{array}\right), \tag{2}$$

   where we used the local ordering again, (1e, 1h, 2e, 2h, 3e, 3h, ...).

3. Just by inspection, it is clear that there is a zero-eigenvalue eigenvector of this $\mathcal{H}$: with

$$\psi_{1+} = \frac{1}{\sqrt{2}} \begin{pmatrix} 1 \\ 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ \dots \end{pmatrix}, \tag{3}$$

   we have $\mathcal{H}\psi_{1+} = 0 = 0 \cdot \psi_{1+}$. Hence, we call ψ_{1+} a *zero-energy mode* or just *zero mode* for short.

4. Remarks:

   (a) ψ_{1+} is its own particle-hole partner: $\sigma_x K \psi_{1+} = \psi_{1+}$. Note that this is possible because it is a zero mode.

(b) The Fock-space excitation operator corresponding to ψ_{1+} is not fermionic: $d^\dagger_{1+} = (c^\dagger_1 + c_1)/\sqrt{2}$. This operator is self-adjoint: $d^\dagger_{1+} = d_{1+}$. Therefore, certain fermionic anticommutator relations are violated: $\{d_{1+}, d_{1+}\} = \{d^\dagger_{1+}, d^\dagger_{1+}\} = 1 \neq 0$, whereas for fermionic operators these would be zero.

(c) The BdG eigenvector ψ_{1+} is an example of *Majorana zero modes* (MZM). In general, an eigenvector of a BdG Hamiltonian is called a MZM, if it is a localized zero mode that is its own particle-hole partner.

(d) *Robustness.* The key observation is as follows. Even if the Hamiltonian is deformed adiabatically away from the fully dimerized topological limit, the MZM survives: it can get deformed, but its zero-energy nature, its localized character, and its Majorana property are all robust.

This is true, because as long as the bulk gap is not closed, its energy of the state developing from ψ_{1+} cannot move away from zero, since it is its on particle-hole partner, and the built-in particle-hole symmetry of the BdG description ensures an up-down symmetric BdG spectrum.

Importantly, this argument does not rely on *homogeneity*. Therefore, it applies also in the presence of disorder. In contrast to the SSH model, where the zero-mode edge states are robust against chiral-symmetric disorder, the edge-localized MZMs of the Kitaev chain are protected against any type of disorder as long as the Fock-space Hamiltonian is quadratic. This robustness is a strong hint that one could 'hide' quantum information against disorder and noise in quantum bits based MZMs.

Henceforth, we will use χ to denote the MZM BdG wave function that developes from ψ_{1+} as the topological fully dimerized limit is left behind.

By definition, as long as the deformation of the Hamiltonian is adiabatic, the bulk gap remains open. This implies that the MZM χ must have an exponentially decaying wave function as the function of the distance from the edge. We'll denote the corresponding decay length as ξ.

(e) We note that since the previous argument relies on particle-hole symmetry, therefore it does not speak about the situation when non-quadratic terms are introduced adiabatically in the Hamiltonian; for example, four-operator products describing electron-electron interaction. For those cases, we cannot conclude that the MZM is robust against adiabatic deformations.

5. From the previous considerations on a half-infinite Kitaev chain, we can draw the following conclusions for a finite Kitaev chain:

(a) MZMs should be robust, as long as the chain is much longer than the decay length of the MZMs on the two edges, that is, if $\xi \ll N$.

(b) For a finite chain in the topological fully dimerized limit, there is a MZM ψ_{1+} at the left edge, and there is another MZM ψ_{N-} at the right edge:

$$\psi_{N-} = \frac{1}{\sqrt{2}} \begin{pmatrix} \dots \\ 0 \\ 0 \\ i \\ -i \end{pmatrix}. \tag{4}$$

Note that $(-1)\psi_{1+}$ and $(-1)\psi_{N-}$ are also MZMs, but in general, multiplying these with an arbitrary global phase factor can destroy their Majorana character, by making them different from their particle-hole partner.

Note an apparent paradox here: in Lecture 1, when discussing the BdG trick, we promised that for an N-site chain, there will be N positive-energy BdG modes that correspond to fermionic quasiparticles. Now, apparently, we've found 2 MZMs, which are not fermionic. The resolution of the apparent paradox is as follows. We can define a fermionic quasiparticle from the two MZMs:

$$\hat{d}_M = \frac{1}{\sqrt{2}} \left(\hat{\gamma}_L + i\hat{\gamma}_R \right), \tag{5}$$

where γ_L and γ_R are the non-fermionic excitations associated to the two MZMs χ_L and χ_R, respectively.

Since both $\hat{\gamma}_L$ and $\hat{\gamma}_R$ are zero-energy excitations, the above considerations imply that *a sufficiently long topological* (not necessarily fully dimerized, can even be disordered) *Kitaev chain has a twofold degenerate ground state*: if $|G\rangle$ is the ground state which is annihilated by all fermionic excitation operators, including the relation $\hat{d}_M |G\rangle = 0$, then $\hat{d}^\dagger_M |G\rangle$ is another state which is orthogonal to $|G\rangle$ but has the same energy as $|G\rangle$. Note that the two ground states $|G\rangle$ and $\hat{d}^\dagger_M |G\rangle$ have opposite fermion number parities, one is even, the other is odd.

II. A NUMERICAL DEMONSTRATION

The robustness of the MZMs is demonstrated by the numerical results attached to these notes. Thanks for Peter Boross for sharing these results.

Here, we highlight the most important features.

A. Definitions

This is the Mathematica code used to obtain the results

B. BdG Hamiltonian

This is the 20×20 BdG Hamiltonian of a Kitaev chain. All matrix elements are independently tuned - this feature is used later when a disordered chain is considered.

C. Fully dimerized

1. This block shows the results corresponding to the topological fully dimerized limit of the 10-site Kitaev chain.

2. The spectrum on the left shows the bulk states away from zero energy as black points, and the two edge-localized states at zero energy as the red and blue points.

3. The numerical diagonalizer actually (incidentally) gives the Majorana zero modes as the BdG eigenvectors, and not some linear combinations of them, which would also be valid zero modes. This is seen in the four subfigures under "Majorana basis". Those are converted by a unitary transformation (see 4th line of code above figures) to bond-antibonding type combinations, which represent a fermionic excitations and its hermitian conjugate (aka its particle-hole partner), shown under the label "fermionic basis".

D. Finite detuning

1. This shows the case when the system is detuned from the topological fully dimerized limit - the first line of code defines $v = 1$ and $\Delta = 0.5$.

2. The bulk states away from zero energy do not any more form a flat ensemble.

3. The blue and red zero modes are still at zero energy, even though the system has been detuned from the topological limit.

4. The apparent zero modes are actually not exactly zero modes: there is an invisibly small minigap between the red and blue points. As a consequence, the ambiguity of choosing the zero-mode representation is gone, and the numerical diagonalizer gives fermionic BdG wave functions.

E. Disorder

1. Here, we're back to the topological fully dimerized limit, but on top of that, disorder in the chemical potential is switched on: For each of the 10 sites, the on-site energy is drawn from a Gaussian ensemble, with zero mean and standard deviation $\sigma_\mu = 0.5$, as defined in the first line of the code.

2. The spectrum reveals that the zero modes are still there, despite disorder.

3. The BdG wave functions are still localized, although their left-right symmetry is gone, due to disorder.

4. The scatter plot serves to further illustrate the robustness of the zero modes, and hence the robustness of the ground-state degeneracy of the Kitaev wire. Here, disorder is introduced in each of the 3 parameters: hopping ($v \mapsto 1 + \delta v$), pair potential ($\Delta \mapsto 1 + \delta\Delta$), and chemical potential ($\mu \mapsto \delta\mu$), and these random contributions δv, $\delta\Delta$ and $\delta\mu$ are generated as independent Gaussian random variables with zero mean and varying standard deviation σ.

5. For each value of σ, the scatter plot shows the full BdG spectrum (20 energy levels) for one random realization of the BdG Hamiltonian. A BdG eigenvalue is colored red, if its eigenstate is an edge state, defined as having more than 75% of its weight on the leftmost and rightmost sites together.

6. The scatter plot shows, up to a significant level of disorder $\sigma \lesssim 0.25$, that the red-colored zero eigenvalues are hardly influenced by disorder: they remain localized (red) and their energy remains very close to zero. This behavior is in stark contrast to the black bulk states, whose energies get randomized in a pace that is roughly linear in the disorder strength.

7. The last subfigure illustrates that the minigap between the apparently zero-energy modes is an exponentially decaying function of the chain length N. This figure is made in the topological fully dimerized limit, with disorder of strength 0.1 on all three parameters. The points denote the average minigaps, averaged over 1000 random realizations. The points show an approximately exponential trend. [The lower (higher) error bars correspond to first (ninth) decile, respectively - see `https://en.wikipedia.org/wiki/Decile`.]

Definitions

In[9]:=

```
Unprotect[Conjugate];
Conjugate /: MakeBoxes[Conjugate[x_], StandardForm] := TemplateBox[
  {Parenthesize[x, StandardForm, Power]},
  "Conjugate",
  DisplayFunction → (SuperscriptBox[#1, "*"] &)]
Protect[Conjugate];

MakeBoxes[Abs[x_], StandardForm] := MakeBoxes@BracketingBar[x]

H[N_] := DiagonalMatrix[Flatten@Table[{-μ_i, μ_i}, {i, N}], 0, {2 N, 2 N}] +
   DiagonalMatrix[Flatten@Table[{0, -Conjugate[Δ_i,i+1]}, {i, N}], 1, {2 N, 2 N}] +
   DiagonalMatrix[Flatten@Table[{0, -Δ_i,i+1}, {i, N}], -1, {2 N, 2 N}] +
   DiagonalMatrix[Flatten@Table[{v_i,i+1, -Conjugate[v_i,i+1]}, {i, N}], 2, {2 N, 2 N}] +
   DiagonalMatrix[Flatten@Table[{Conjugate[v_i,i+1], -v_i,i+1}, {i, N}], -2, {2 N, 2 N}] +
   DiagonalMatrix[Flatten@Table[{Δ_i,i+1, 0}, {i, N}], 3, {2 N, 2 N}] +
   DiagonalMatrix[Flatten@Table[{Conjugate[Δ_i,i+1], 0}, {i, N}], -3, {2 N, 2 N}];

H[N_, {v′_, σv_}, {Δ′_, σΔ_}, {μ′_, σμ_}] := H[N] /. Flatten@Table[{
     v_i,i+1 → v′ + If[σv == 0, 0, RandomVariate[NormalDistribution[0, σv]]],
     Δ_i,i+1 → Δ′ + If[σΔ == 0, 0, RandomVariate[NormalDistribution[0, σΔ]]],
     μ_i → μ′ + If[σμ == 0, 0, RandomVariate[NormalDistribution[0, σμ]]]}, {i, N}]

PlotOpt = {
   PlotRange → All,
   Frame → True,
   FrameStyle → Black,
   ImageSize → 400,
   LabelStyle → Directive[FontSize → 14, FontFamily → "Gill Sans", FontColor → Black]
  };

PlotOpt′ = {
   PlotRange → All,
   PlotMarkers → Automatic,
   Frame → True,
   FrameStyle → Black,
   ImageSize → 200,
   LabelStyle → Directive[FontSize → 14, FontFamily → "Gill Sans", FontColor → Black]
  };
```

BdG Hamiltonian

In[17]:=

```
N = 10; Clear[v]; Clear[Δ]; Clear[μ];
Grid[H[N], Frame → {None, None, Flatten[Table[{{2 k - 1, 2 k}, {2 l - 1, 2 l}} → True, {k, N}, {l, N}], 1]}]
```

Out[18]=

$-\mu_1$	0	$v_{1,2}$	$\Delta_{1,2}$	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0	μ_1	$-\Delta_{1,2}^*$	$-v_{1,2}^*$	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
$v_{1,2}^*$	$-\Delta_{1,2}$	$-\mu_2$	0	$v_{2,3}$	$\Delta_{2,3}$	0	0	0	0	0	0	0	0	0	0	0	0	0	0
$\Delta_{1,2}^*$	$-v_{1,2}$	0	μ_2	$-\Delta_{2,3}^*$	$-v_{2,3}^*$	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0	0	$v_{2,3}^*$	$-\Delta_{2,3}$	$-\mu_3$	0	$v_{3,4}$	$\Delta_{3,4}$	0	0	0	0	0	0	0	0	0	0	0	0
0	0	$\Delta_{2,3}^*$	$-v_{2,3}$	0	μ_3	$-\Delta_{3,4}^*$	$-v_{3,4}^*$	0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	$v_{3,4}^*$	$-\Delta_{3,4}$	$-\mu_4$	0	$v_{4,5}$	$\Delta_{4,5}$	0	0	0	0	0	0	0	0	0	0
0	0	0	0	$\Delta_{3,4}^*$	$-v_{3,4}$	0	μ_4	$-\Delta_{4,5}^*$	$-v_{4,5}^*$	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	$v_{4,5}^*$	$-\Delta_{4,5}$	$-\mu_5$	0	$v_{5,6}$	$\Delta_{5,6}$	0	0	0	0	0	0	0	0
0	0	0	0	0	0	$\Delta_{4,5}^*$	$-v_{4,5}$	0	μ_5	$-\Delta_{5,6}^*$	$-v_{5,6}^*$	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	$v_{5,6}^*$	$-\Delta_{5,6}$	$-\mu_6$	0	$v_{6,7}$	$\Delta_{6,7}$	0	0	0	0	0	0
0	0	0	0	0	0	0	0	$\Delta_{5,6}^*$	$-v_{5,6}$	0	μ_6	$-\Delta_{6,7}^*$	$-v_{6,7}^*$	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	$v_{6,7}^*$	$-\Delta_{6,7}$	$-\mu_7$	0	$v_{7,8}$	$\Delta_{7,8}$	0	0	0	0
0	0	0	0	0	0	0	0	0	0	$\Delta_{6,7}^*$	$-v_{6,7}$	0	μ_7	$-\Delta_{7,8}^*$	$-v_{7,8}^*$	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0	$v_{7,8}^*$	$-\Delta_{7,8}$	$-\mu_8$	0	$v_{8,9}$	$\Delta_{8,9}$	0	0
0	0	0	0	0	0	0	0	0	0	0	0	$\Delta_{7,8}^*$	$-v_{7,8}$	0	μ_8	$-\Delta_{8,9}^*$	$-v_{8,9}^*$	0	0
0	0	0	0	0	0	0	0	0	0	0	0	0	0	$v_{8,9}^*$	$-\Delta_{8,9}$	$-\mu_9$	0	$v_{9,10}$	$\Delta_{9,10}$
0	0	0	0	0	0	0	0	0	0	0	0	0	0	$\Delta_{8,9}^*$	$-v_{8,9}$	0	μ_9	$-\Delta_{9,10}^*$	$-v_{9,10}^*$
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	$v_{9,10}^*$	$-\Delta_{9,10}$	$-\mu_{10}$	0
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	$\Delta_{9,10}^*$	$-v_{9,10}$	0	μ_{10}

Fully dimerized

In[19]:=

```
N = 10; v = 1; σv = 0; Δ = 1; σΔ = 0; μ = 0; σμ = 0;
Eψs = SortBy[Transpose[Chop[Eigensystem[N@H[N, {v, σv}, {Δ, σΔ}, {μ, σμ}]]]], First];
{ψ1, ψ2} = MinimalBy[Eψs, Abs[#[[1]]] &, 2][[All, 2]];
ψ1′ = (ψ1 + ⅈ ψ2)/√2; ψ2′ = (ψ1 - ⅈ ψ2)/√2;

Grid[{{"", Text[Style["Majorana basis", FontFamily → "Gill Sans"]], Text[Style["Fermionic basis", FontFamily → "Gill Sans"]]},
  {ListPlot[{#[[1 ;; N - 1]], {#[[N]]}, {#[[N + 1]]}, #[[N + 2 ;; -1]]} &@Transpose[{Range[2 N], Eψs[[All, 1]]}], PlotOpt, FrameLabel → {"#state", "E/v"},
    PlotStyle → {Black, Red, Blue, Black}], Grid[{{
      ListLinePlot[Abs[ψ1[[1 ;; -1 ;; 2]]]^2, PlotOpt′, FrameLabel → {"#site", "|u|^2"}, PlotStyle → Red],
      ListLinePlot[Abs[ψ1[[2 ;; -1 ;; 2]]]^2, PlotOpt′, FrameLabel → {"#site", "|v|^2"}, PlotStyle → Red]}, {
      ListLinePlot[Abs[ψ2[[1 ;; -1 ;; 2]]]^2, PlotOpt′, FrameLabel → {"#site", "|u|^2"}, PlotStyle → Blue],
      ListLinePlot[Abs[ψ2[[2 ;; -1 ;; 2]]]^2, PlotOpt′, FrameLabel → {"#site", "|v|^2"}, PlotStyle → Blue]}}],
   Grid[{{
      ListLinePlot[Abs[ψ1′[[1 ;; -1 ;; 2]]]^2, PlotOpt′, FrameLabel → {"#site", "|u|^2"}, PlotStyle → Black],
      ListLinePlot[Abs[ψ1′[[2 ;; -1 ;; 2]]]^2, PlotOpt′, FrameLabel → {"#site", "|v|^2"}, PlotStyle → Black]}, {
      ListLinePlot[Abs[ψ2′[[1 ;; -1 ;; 2]]]^2, PlotOpt′, FrameLabel → {"#site", "|u|^2"}, PlotStyle → Black],
      ListLinePlot[Abs[ψ2′[[2 ;; -1 ;; 2]]]^2, PlotOpt′, FrameLabel → {"#site", "|v|^2"}, PlotStyle → Black]}}]}}]
```

Out[23]=



Finite detuning

In[24]:=

```
N = 10; v = 1; σv = 0; Δ = 0.5; σΔ = 0; μ = 0; σμ = 0;
Eψs = SortBy[Transpose[Chop[Eigensystem[N@H[N, {v, σv}, {Δ, σΔ}, {μ, σμ}]]]], First];
{ψ1, ψ2} = MinimalBy[Eψs, Abs[#[[1]]] &, 2][[All, 2]];

Grid[{{"", Text[Style["Fermionic basis", FontFamily → "Gill Sans"]]},
  {ListPlot[{#[[1 ;; N - 1]], {#[[N]]}, {#[[N + 1]]}, #[[N + 2 ;; -1]]} &@Transpose[{Range[2 N], Eψs[[All, 1]]}], PlotOpt, FrameLabel → {"#state", "E/v"},
    PlotStyle → {Black, Red, Blue, Black}], Grid[{{
      ListLinePlot[Abs[ψ1[[1 ;; -1 ;; 2]]]^2, PlotOpt, FrameLabel → {"#site", "|u|^2"}, PlotStyle → Red],
      ListLinePlot[Abs[ψ1[[2 ;; -1 ;; 2]]]^2, PlotOpt, FrameLabel → {"#site", "|v|^2"}, PlotStyle → Red]}, {
      ListLinePlot[Abs[ψ2[[1 ;; -1 ;; 2]]]^2, PlotOpt, FrameLabel → {"#site", "|u|^2"}, PlotStyle → Blue],
      ListLinePlot[Abs[ψ2[[2 ;; -1 ;; 2]]]^2, PlotOpt, FrameLabel → {"#site", "|v|^2"}, PlotStyle → Blue]}}]}}]
```

Out[27]=



Disorder

In[28]:=

```
N = 10; v = 1; σv = 0; Δ = 1; σΔ = 0; μ = 0; σμ = 0.5;
Eψs = SortBy[Transpose[Chop[Eigensystem[N@H[N, {v, σv}, {Δ, σΔ}, {μ, σμ}]]]], First];
{ψ1, ψ2} = MinimalBy[Eψs, Abs[#⟦1⟧] &, 2]⟦All, 2⟧;

Grid[{{"", Text[Style["Fermionic basis", FontFamily → "Gill Sans"]]},
  {ListPlot[{#⟦1 ;; N - 1⟧, {#⟦N⟧}, {#⟦N + 1⟧}, #⟦N + 2 ;; -1⟧} &@Transpose[{Range[2 N], Eψs⟦All, 1⟧}], PlotOpt, FrameLabel → {"♯state", "E/v"},
    PlotStyle → {Black, Red, Blue, Black}], Grid[{{
      ListLinePlot[Abs[ψ1⟦1 ;; -1 ;; 2⟧]^2, PlotOpt, FrameLabel → {"♯site", "|u|^2"}, PlotStyle → Red],
      ListLinePlot[Abs[ψ1⟦2 ;; -1 ;; 2⟧]^2, PlotOpt, FrameLabel → {"♯site", "|v|^2"}, PlotStyle → Red]}, {
      ListLinePlot[Abs[ψ2⟦1 ;; -1 ;; 2⟧]^2, PlotOpt, FrameLabel → {"♯site", "|u|^2"}, PlotStyle → Blue],
      ListLinePlot[Abs[ψ2⟦2 ;; -1 ;; 2⟧]^2, PlotOpt, FrameLabel → {"♯site", "|v|^2"}, PlotStyle → Blue]}}]}}]
```

Out[31]=

Fermionic basis



In[32]:=

```
N = 10; v = 1; Δ = 1; μ = 0;
data = Flatten[Table[Transpose[Prepend[Chop[Eigensystem[N@H[N, {v, σ}, {Δ, σ}, {μ, σ}]]], ConstantArray[σ, 2 N]]], {σ, 0, 1, 0.005}], 1];
data = GatherBy[data, Total[Abs[#⟦-1, {1, 2, 2 N - 1, 2 N}⟧]^2] > 0.75 &];
ListPlot[data⟦All, All, {1, 2}⟧, PlotStyle → {Black, Red}, PlotOpt, FrameLabel → {"σ/v", "E/v"}, PlotMarkers → {Automatic, 2.9}]
```

Out[34]=



In[35]:=

```
v = 1; σv = 0.1; Δ = 1; σΔ = 0.1; μ = 0; σμ = 0.1; Nr = 1000;
data = Table[{N, Table[Min[Abs[Eigenvalues[H[N, {v, σv}, {Δ, σΔ}, {μ, σμ}]]]], Nr]}, {N, 2, 10}];
ListLogPlot[Map[{#⟦1⟧, Around[Mean[#⟦2⟧], {Mean[#⟦2⟧] - Quantile[#⟦2⟧, 0.1], Quantile[#⟦2⟧, 0.9] - Mean[#⟦2⟧]}]} &, data], PlotRange → Automatic,
 PlotOpt, FrameLabel → {"N", "minigap/v"}, PlotStyle → Black]
```

Out[37]=